
Методы математической физики
Введение

По каким функциям разлагать решение в
произвольной области?

В случае круга и прямоугольника было вполне понятно, по
каким функциям вести разложение. Для произвольной области
указать подходящий набор функций удалось лишь в 1940-х
годах, когда появился абрисс того, что теперь называют
методом конечных элементов (МКЭ, FEM) и открытие которого
обычно связывают с именем Рихарда Куранта (Courant, 1943).
Название – метод конечных элементов – было предложено
Клафом в середине 1960-х годов [1].

FEA Software

Работа Куранта не была замечена, но в конце 1950-х годах
инженеры, занимавшиеся сопроматом, переоткрыли МКЭ.
Рост популярности МКЭ во многом объясняется удобством его
реализации на персональном компьютере. В настоящее время
имеется великое множество компьютерных систем
конечно-элементного анализа (FEA Software).

ANSYS, Inc.

В 1970 г. Джон Свенсон (John
A. Swanson) в Пенсельвании
основал Swanson Analysis
Systems, Inc. (SASI),
ставящую перед собой цель
развитие программного
обеспечение, основанного на
МКЭ. В 1994 г. компания
была продана и
переименована в ANSYS, Inc.;
одноименный продукт до сих
пор существует на рынке.

YouTube

http://www.ansis.com

http://www.youtube.com/watch?v=kfFpI2uK6rQ&list=UUyUjgCTvvFuzAcZf3IITffg&index=3&feature=plcp
http://www.ansis.com

MatLab

В состав стандартной
поставки MatLab’а входит Pde
Toolbox, умеющий решать
МКЭ основные линейные
задачи мат. физики в плоских
областях.
Хорошо интегрируется в
MatLab пакет, созданный
Comsol Multiphysics; его
обычно называют FEMLab’ом.

YouTube

http://www.comsol.com

http://www.comsol.com
http://www.youtube.com/watch?v=GcL9iNVeykA
http://www.comsol.com

FreeFem++

Мы будем использовать бесплатную программу FreeFem++,
созданную в Лаборатории Ж.-Л. Лионса в Париже и
поддерживающую большое число ОС: Linux, Mac OS X,
Microsoft Windows и Solaris. Входит в репозиторий Ubuntu,
начиная с версии 12.04.
Для справок см.:

Hecht F. FreeFem++. Third Edition, Version 3.19. Laboratoire
Jacques-Louis Lions, Université Pierre et Marie Curie, Paris,
2012.
Жуков М.Ю., Ширяева Е.В. Пакет конечных элементов
FreeFem++. Ростов-на-Дону, 2010

http://www.freefem.org

http://www.freefem.org
http://www.freefem.org

Задача

Первая краевая задача для
уравнения Пуассона.

Уравнение Пуассона

Простейшая краевая задача в 2d

Простейшей с точки зрения МКЭ является первая краевая
задача для уравнения Пуассона, а именно{︂

Δ𝑢 = 𝑓, (𝑥, 𝑦) ∈ Ω
𝑢|𝜕Ω = 0.

(1)

Будем считать, что Ω – односвязная область на плоскости,
ограниченная замкнутой кривой.

Классическое решение

Определение
Примем, что классическое решение дважды непрерывно
дифференцируемо в области Ω и на ее границе.

Единственность такого решения очевидна: разность двух
классических решений является решением однородной первой
краевой задачи для уравнения Лапласа, и следовательно, равно
нулю.
Вопрос о существовании классического решения оставим на
потом, сосредоточившись на конструктивном построении
такового.

Метод конечных элементов

Применение МКЭ к любой линейной краевой задачи мат.
физики удобно разбить на несколько шагов:

1 триангуляция области,
2 построение системы базисных функций,
3 дискретизация дифференциального уравнения,
4 решение системы линейных алгебраически уравнений.

Шаг № 1

Триангуляция области

Триангуляция области

Разделим область Ω на треугольники, имеющие сходные
размеры и пропорции. Линейные размеры треугольников сети
характеризуют числом ℎ, равным максимальной длине сторон
треугольников.
В ручную заданную на клетчатой бумаге область поделить на
небольшое число примерно равносторонних треугольников
совсем не трудно. Если же треугольников много, применяют
специальные алгоритмы триангуляции, восходящие к работе
Б.Н. Делоне 1934 г. [4]

Триангуляция области

Терминология

Сами треугольники и называют конечными элементами, их
вершины делят на внутренние, т.е. лежащие внутри области Ω,
и граничные, то есть лежащие на границе этой области.
Две вершины будем называть соседними, если они являются
вершинами одного треугольника.
Перенумеруем вершины в таком порядке, чтобы первыми шли
внутренние вершины, а затем граничные. Число внутренних
вершин будем обозначать как 𝑁 .

Вершины

𝐴1, . . . 𝐴6,
граничные:

соседние с 𝐴1:
𝐴2, 𝐴3, 𝐴4 и
𝐴7, 𝐴8, 𝐴9.

Пример: триангуляция эллипса

Для триангуляции эллипса создадим файл след. содержания.

Файл Poisson-1.edp
border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);}
mesh Th = buildmesh (C(20));
plot(Th, ps="Poisson-1.eps",value=true,fill=false);

Здесь в первой строке задается граница 𝐶 эллипса в
параметрической форме, во второй задается сеть 𝑇ℎ с 20-тью
вершинами на границе 𝐶, в третьей указано, что результат
надо записать в графический файл Poisson-1.eps.

Пример: триангуляция эллипса

Теперь откроем терминал, перейдем в директорию, в которой
находится файл Poisson-1.edp, и выполним команду

Терминал.
FreeFem++-nw Poisson-1.edp

В результате в той же директории будет создан файл
Poisson-1.eps.

Замечание. – Использование в пути к файлу русских букв может
стать источником проблем.

Результат: файл Poisson-1.eps

Шаг № 2

Построение системы базисных
функций

Базисные функции

С каждой вершиной 𝐴𝑖 свяжем функцию 𝜙𝑖, которая линейна
во всех конечных элементах и 𝜙𝑖(𝐴𝑗) = 𝛿𝑖𝑗 . Эта функция будет
отлична от нуля только в тех треугольниках, одной из вершин
которых является точка 𝐴𝑖.

График базисной функции 𝜙𝑖

График функции 𝜙𝑖

представляет собой
пирамиду высоты 1,
основанием которой
служит полигон с
вершинами в точках,
соседних с точкой 𝐴𝑖,
а над самой точкой 𝐴𝑖

размещается вершина
пирамиды.

Линейная комбинация базисных функций

Линейная комбинация этих функций∑︁
𝑖

𝑔𝑖𝜙𝑖(𝑥, 𝑦)

является кусочно-линейной функцией, причем коэффициенты
этой комбинации имеют очень простой геометрический смысл:
𝑔𝑖 – значение этой функции в вершине 𝐴𝑖.

Разложение по базису

Для произвольной функции 𝑔 разность

𝑔(𝑥, 𝑦)−
∑︁
𝑖

𝑔(𝑥𝑖, 𝑦𝑖)𝜙𝑖(𝑥, 𝑦)

равна нулю во всех вершинах триангуляции.
Поскольку вершины триангуляции расположены друг от друга
на расстоянии, примерно равном ℎ, при надлежащих
ограничениях на рост производной этой функции, эта функция
всюду не больше ℎ.

Разложение по базису

Гипотеза.
Любую функцию 𝑔 можно аппроксимировать линейной
комбинацией базисных функций, то есть с ошибкой порядка ℎ
верно

𝑔(𝑥, 𝑦) ≈
∑︁
𝑖

𝑔(𝑥𝑖, 𝑦𝑖)𝜙𝑖(𝑥, 𝑦).

Справедливость этой гипотезы подтверждают многочисленные
примеры.

Пример: аппроксимация функции 𝑔 = exp(−𝑥2 − 𝑦2).

IsoValue
0.127436
0.171509
0.215583
0.259656
0.303729
0.347803
0.391876
0.435949
0.480022
0.524096
0.568169
0.612242
0.656316
0.700389
0.744462
0.788536
0.832609
0.876682
0.920756
0.964829

Пример

Для построения этого рисунка в FreeFem++ нужно создать и
выполнить файл след. содержания.

Файл Poisson-2.edp
border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);}
mesh Th = buildmesh (C(20));
fespace Vh(Th,P1);
Vh gh=exp(-x^2-y^2);
plot(Th,gh, ps="Poisson-2.eps",value=true,fill=false);

Здесь в третьей строке задается линейная оболочка Vℎ,
натянутая на базисные функции, т.н. пространство конечных
элементов (f. e. space), в четвертой вычисляется
аппроксимация 𝑔ℎ для функция exp(−𝑥2 − 𝑦2).

Замечание

При изменении в 3-ей строке параметра P1, скажем, на P2,
вместо аппроксимации линейными функциями получится
аппроксимация полиномами второго порядка.

Шаг № 3

Дискретизация уравнения
Пуассона.

Приближенное решение

Будем искать решение краевой задачи (1) в виде линейной
комбинации базисных функций, отвечающих внутренним
вершинам:

𝑢 =

𝑁∑︁
𝑖=1

𝑢𝑖𝜙𝑖(𝑥, 𝑦);

число 𝑁 при этом именуют числом свободных параметров. Мы
берем в этой сумме только базисные функции, отвечающие
внутренним вершинам, поскольку на границе решение должно
обращаться в нуль.
Мы не можем подставить это выражение в уравнение Пуассона
Δ𝑢 = 𝑓 , поскольку базисные функции не имеют второй
производной.

Слабая постановка задачи

Умножим уравнение Пуассона Δ𝑢 = 𝑓 на произвольную
функцию 𝑣 и проинтегрируем по области. В силу формулы
Гаусса-Остроградского тождество∫︁

Ω

𝑑𝑥𝑑𝑦(∇𝑢,∇𝑣) +

∫︁
Ω

𝑑𝑥𝑑𝑦𝑓𝑣 = 0 (2)

верно для любой функции 𝑣, лишь бы она была непрерывна в
Ω, кусочно-дифференцируема функций в Ω и равна нулю на
границе.

СЛАУ

В частности при 𝑣 = 𝜙𝑗 для точного решения 𝑢 верно∫︁
Ω

𝑑𝑥𝑑𝑦(∇𝑢,∇𝜙𝑗) +

∫︁
Ω

𝑑𝑥𝑑𝑦𝑓𝜙𝑗 = 0. (𝑗 = 1, 2, . . . 𝑁)

Подставляя сюда

𝑢 =

𝑁∑︁
𝑖=1

𝑢𝑖𝜙𝑖(𝑥, 𝑦)

получим систему из 𝑁 алгебраических уравнений для
отыскания 𝑁 коэффициентов 𝑢𝑖, именно

𝑁∑︁
𝑖=1

∫︁
Ω

𝑑𝑥𝑑𝑦(∇𝜙𝑖,∇𝜙𝑗)𝑢𝑖 +

∫︁
Ω

𝑑𝑥𝑑𝑦𝑓𝜙𝑗 = 0 (𝑗 = 1, 2, . . . 𝑁)



Методы математической физики
Шаг № 3

Приближенное решение

Определение
Мы принимаем за приближенное решение 𝑢ℎ исходной краевой
задачи линейную комбинацию

𝑢ℎ =

𝑁∑︁
𝑖=1

𝑢𝑖𝜙𝑖(𝑥, 𝑦),

которая точно удовлетворят соотношению (2), т.е.∫︁
Ω

𝑑𝑥𝑑𝑦(∇𝑢,∇𝑣) +

∫︁
Ω

𝑑𝑥𝑑𝑦𝑓𝑣 = 0

для всех функций 𝑣, которые сами являются линейной
комбинацией базисных функций.

Шаг № 4

Решение СЛАУ.

СЛАУ в матричной форме

Исходная краевая задача свелась к системе алгебраических
уравнений

Au+ f = 0,

где A – матрица 𝑁 ×𝑁 с элементами

𝑎𝑖,𝑗 =

∫︁
Ω

𝑑𝑥𝑑𝑦(∇𝜙𝑖,∇𝜙𝑗),

f – столбец длины 𝑁 с элементами

𝑓𝑖 =

∫︁
Ω

𝑑𝑥𝑑𝑦𝑓𝜙𝑖,

а u – столбец длины 𝑁 , элементами которого служат искомые
коэффициенты 𝑢𝑖.

СЛАУ большого порядка

Остается получившуюся систему линейных уравнений решить
средствами линейной алгебры. Хотя на практике число
свободных параметров 𝑁 может иметь порядок сотни или даже
тысячи, современные пакеты для работы с матрицами
вычислят u за доли секунды. Однако не следует воображать
себе, что при этом применяется правило Крамера, бывшее
основным способом решения линейных уравнений в учебниках
по Линейной алгебре в XIX веке.
FreeFem++ использует при работе со СЛАУ сторонний пакет –
UMFPACK (unsymmetric multifrontal sparse LU factorization
package) Тима Девиса (Timothy A. Davis, 1994).

http://www.cise.ufl.edu/research/sparse/umfpack/

Структура матрицы и вершины сетки

Ai

Aj

x

z
y

Если 𝑖-ая и
𝑗-ая вершины
не являются
соседними, то
𝑎𝑖𝑗 = 𝑎𝑗𝑖 = 0.

Разреженность матрицы

Если каждая вершина
соседствует не более чем с 𝑚
другими вершинами, то среди
𝑁2 элементов матрицы
имеется лишь 𝑚𝑁 ненулевых
элементов. Такие матрицы
называются разреженными и
последние лет 40 развития
линейной алгебры были
направлены на изобретение
особых алгоритмов работы с
такими матрицами [5].

Показать

http://www.cise.ufl.edu/~davis/Harn/

Ответ

Остается написать ответ или в виде суммы

𝑢 =

𝑁∑︁
𝑖=1

𝑢𝑖𝜙𝑖(𝑥, 𝑦),

или приняв 𝑢𝑖 за найденные значения решения в вершинах
{𝐴𝑖}.
Замечание. – Как отмечалось выше, часто интересно знать не
саму функцию 𝑢, а ее максимум. Для его отыскания больше не
надо ничего дифференцировать, просто

𝑢max = max
𝑖=1..𝑁

𝑢𝑖.

Пример.

Краевая задача в эллипсе.

Постановка задачи.

Рассмотрим краевую задачу{︁
Δ𝑢+ 𝑒−𝑥2−𝑦2 = 0, 𝑢|𝐶 = 0

в уже знакомом нам эллипсе 𝐶. Специфической особенностью
FreeFem++ является необходимость задавать уравнение в
слабой форме (2), т.е. в данном случае∫︁

Ω

𝑑𝑥𝑑𝑦(∇𝑢,∇𝑣)−
∫︁
Ω

𝑑𝑥𝑑𝑦𝑒−𝑥2−𝑦2𝑣 = 0

для любой функции 𝑣, обращающейся в нуль на границе 𝐶.

Пример

Решение задачи в FreeFem++.

Файл Poisson-3.edp
border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);}
mesh Th = buildmesh (C(20));
fespace Vh(Th,P1);
Vh u,v;
solve Poisson(u,v,solver=LU) =
int2d(Th)(dx(u)*dx(v) + dy(u)*dy(v))
+ int2d(Th)( -exp(-x^2-y^2)*v)
+ on(C,u=0);

plot(Th,u,ps="Poisson-3.eps",value=true,fill=false);

Ответ: файл Poisson-3.eps

IsoValue
0.00616232
0.018487
0.0308116
0.0431362
0.0554609
0.0677855
0.0801102
0.0924348
0.104759
0.117084
0.129409
0.141733
0.154058
0.166383
0.178707
0.191032
0.203357
0.215681
0.228006
0.24033

Какую СЛАУ решает FreeFem++ на самом деле?

Для обозримости возьмем не
20, а всего 6 точек на
границе. По идеи, в этом
случае имеется 1 базовая
функция и, следовательно,
СЛАУ сводится к одному
уравнению.

IsoValue
0.00606886
0.0182066
0.0303443
0.042482
0.0546197
0.0667574
0.0788952
0.0910329
0.103171
0.115308
0.127446
0.139584
0.151721
0.163859
0.175997
0.188135
0.200272
0.21241
0.224548
0.236685

Пример

Вывод СЛАУ на печать

Файл Poisson-4.edp
border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);}
mesh Th = buildmesh (C(6));
fespace Vh(Th,P1);
Vh u,v;
varf a(u,v) = int2d(Th)( dx(u)*dx(v) + dy(u)*dy(v))
+ on(C,u=0) ;
matrix A=a(Vh,Vh);
varf l(unused,v) = int2d(Th)(exp(-x^2-y^2)*v)+on(C,unused=0);
Vh f; f[] = l(0,Vh);
u[]=A^-1*f[];
cout << "A = " << A << endl;
cout << "f = " << f[] << endl;
cout << "u = " << u[] << endl;
plot(Th,u,ps="Poisson-4.eps",value=true,fill=false);

СЛАУ в FreeFem++.

⎛⎜⎜⎜⎜⎜⎜⎜⎜⎝

1030 −0.4 −0.4 −0.1
−0.4 1030 0 −0.9 −0.1
−0.4 0 1030 −0.9 0 −0.1
−0.1 −0.9 −0.9 3.7 −0.9 −0.9 −0.1
0 −0.1 0 −0.9 1030 0 −0.4
0 0 −0.1 −0.9 0 1030 −0.4
0 0 0 −0.1 −0.4 −0.4 1030

⎞⎟⎟⎟⎟⎟⎟⎟⎟⎠

⎛⎜⎜⎜⎜⎜⎜⎜⎜⎝

𝑢1

𝑢2

𝑢3

𝑢4

𝑢5

𝑢6

𝑢7

⎞⎟⎟⎟⎟⎟⎟⎟⎟⎠
=

⎛⎜⎜⎜⎜⎜⎜⎜⎜⎝

0
0
0
0.9
0
0
0

⎞⎟⎟⎟⎟⎟⎟⎟⎟⎠

Пример

Реализация условий Дирихле в FreeFem++.

FreeFem++ ищет решение в виде

𝑢 =
∑︁

𝑢𝑖𝜙𝑖,

распространив сумму и на грачиные вершины с № 1-3, 5-7, но
увидев +on(C,u=0), для этих номеров заполняет диагональ
матрицы A очень большим числом 1030, а f – нулями.
Разумеется 6 уравнений (при 𝑖 ̸= 4) дают

1030𝑢𝑖 + · · · = 0 ⇒ 𝑢𝑖 = 0,

а оставшееся

3.7𝑢4 = 0.9 ⇒ 𝑢4 = 0.24.

tgv в FreeFem++.

В документации ([2], стр. 16) это очень большое число так и
называется tgv (très grande valeur), по умолчанию оно равно
1030, но при желании его можно переопределить.
Такой неожиданный способ реализации условий Дирихле был
избран из-за того, что разработчики не захотели писать два
разных блока для условий Дирихле и условий Неймана. При
решении задач с небольшим числом элементов он, конечно,
весьма неэкономичен, а в задачах на собственные значения и
вовсе приводит к появлению лишних собственных значений.

Домашняя работа

Выполненную домашнюю работу следует собрать в один
pdf-файл и послать по адресу mmph@narod.ru, указав в теме
письма номер группы. В ответ придут замечания и
комментарии. После одной такой итерации работа будет
опубликована на сайте http://mmph.narod.ru.

mailto:mmph@narod.ru
http://mmph.narod.ru

Домашняя задача № 1

Используя любую понравившуюся систему
конечно-элементного анализа, решите краевую задачу в круге:{︂

Δ𝑢 = sin𝑥 sin 𝑦, (𝑟 < 1)
𝑢|𝑟=1 = 0

Ответ представьте в виде графика. Что меняется с
увеличением числа треугольников?

Домашняя задача № 2

Докажите след. утверждение: если 𝑢 дважды непрерывно
дифференцируема в замыкании области Ω и удовлетворяет
условию (2) при всех гладких 𝑣, обращающихся в нуль на
границе Ω, то она удовлетворяет и уравнению Пуассона
Δ𝑢 = 𝑓 .

border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);} 
mesh Th = buildmesh (C(20)); 
plot(Th, ps="Poisson-1.eps",value=true,fill=false); 



border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);} 
mesh Th = buildmesh (C(20)); 
plot(Th, ps="Poisson-1.eps",value=true,fill=false); 
fespace Nh(Th,P0);
Nh h = hTriangle;
cout << "max. diametr of triangles = " << h[].max << endl;
Nh a = area;
cout << "max. area of triangles = " << a[].max << endl;



border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);} 
mesh Th = buildmesh (C(20)); 
fespace Vh(Th,P1);
Vh gh=exp(-x^2-y^2);
plot(Th,gh, ps="Poisson-2.eps",value=true,fill=false);



border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);} 
mesh Th = buildmesh (C(20)); 
fespace Vh(Th,P1);
Vh u,v;
solve Poisson(u,v,solver=LU) =  
 int2d(Th)(dx(u)*dx(v) + dy(u)*dy(v))   
 + int2d(Th)( -exp(-x^2-y^2)*v)          
 + on(C,u=0);
plot(Th,u,ps="Poisson-3.eps",value=true,fill=false);



border C(t=0,2*pi){x=1.5*cos(t); y=sin(t);} 
mesh Th = buildmesh (C(6)); 
fespace Vh(Th,P1);
Vh u,v;
varf a(u,v) = int2d(Th)( dx(u)*dx(v) + dy(u)*dy(v))
+ on(C,u=0) ;
matrix A=a(Vh,Vh);
varf l(unused,v) = int2d(Th)(exp(-x^2-y^2)*v)+on(C,unused=0);
Vh f; f[] = l(0,Vh);
u[]=A^-1*f[];
cout << "A = " << A << endl;
cout << "f = " << f[] << endl;
cout << "u = " << u[] << endl;
plot(Th,u,ps="Poisson-4.eps",value=true,fill=false);


